Unit Analysis

Units using Quantity[]

```
In[ ]:= L = Quantity[3, "Meters"]
Out[ ]=
3 m
```

```
In[ ]:= v = Quantity[5, "Inches"/"Seconds"]
Out[ ]=
5 in/s
```

```
In[ ]:= L/v // N
Out[ ]=
23.622 s
```

Units using Free-form Input

Free-form input is typed using ctrl+=
After pressing that combination of keys you should get a little box that you can type in. This box will try to interpret basic sentences or phrases and return what you asked for. For instance typing “thermal conductivity of carbon steel” returns 320. in $BTU_{IT}/(ft^2 h {}^{\circ}_{\Delta} F)$

```
In[ ]:= 5 in/s
        320. in BTU_IT/(ft^2 h °Δ F)
        c
Out[ ]=
5 in/s
Out[ ]=
320. in BTU_IT/(ft^2 h °Δ F)
Out[ ]=
c
```

```
In[ ]:= L/c
Out[ ]=
3 m/c
```

Using UnitConvert[]

In[∘]:=
```
UnitConvert[L/c, "Seconds"] // N
```
Out[∘]=

1.00069×10^{-8} s

In[∘]:=
```
UnitConvert[6 °C, "Kelvins"] // N
```
Out[∘]=

279.15 K

In[∘]:=
```
UnitConvert[6 Δ°C, "KelvinsDifference"]
```
Out[∘]=

6 K

Using QuantityMagnitude[]

In[∘]:=
```
QuantityMagnitude[L/c, "Seconds"] // N
```
Out[∘]=

1.00069×10^{-8}

In[∘]:=
```
(*Replace all units with SI units as normal numbers*)
{3 m, Quantity[5, "Watts"], 2 s} /. a_Quantity :→ QuantityMagnitude[UnitConvert[a]]
```
Out[∘]=

{3, 5, 2}

Independent Units

In[∘]:=
```
Quantity[2, "Hours"/IndependentUnit["Dishwasher Cycle"]]
```
Out[∘]=

2 h/Dishwasher Cycle

What can we actually do with this?

In[∘]:=
```
UnitConvert[Quantity[2, "Hours"/IndependentUnit["Dishwasher Cycle"]] * Quantity[1,
   IndependentUnit["Dishwasher Cycle"]/"Weeks"] * 2000 W * yr, "Kilowatts" "Hours"] // N
```
Out[∘]=

208.571 h kW

Symbolic Replacements

In[ ]:=
```
2 x^2 + 3 x + 1 /. x → 3
```
Out[ ]=

28

In[ ]:=
```
2 x^2 + 3 x + 1 /. x → e^x
```
Out[ ]=

$1 + 3 e^{x} + 2 e^{2x}$

In[ ]:=
```
Plot[Evaluate[2 x^2 + 3 x + 1 /. x → e^x], {x, 0, 10}]
```
Out[ ]=



In[ ]:=
```
∂x Sin[x] /. x → 3
```
Out[ ]=

Cos[3]

Symbolic Simplification

Simple simplification

In[ ]:=
```
Sin[x]^2 + Cos[x]^2
```
Out[ ]=

$\text{Cos}[x]^2 + \text{Sin}[x]^2$

In[ ]:=
```
b[x] + a[x] + c[x]
```
Out[ ]=

a[x] + b[x] + c[x]

In[ ]:=
```
Simplify[Sin[x]^2 + Cos[x]^2]
```
Out[ ]=

1

Simplification with assumptions

In[ ]:=
```
Simplify[Cos[π n] + Sin[π/2 m], Assumptions → {n ∈ Integers, m/2 ∈ Integers}]
Simplify[Cos[π n] + Sin[π/2 m], Assumptions → {n ∈ Integers, m ∈ Integers, Mod[m, 2] == 0}]
```

Out[ ]=

$(-1)^n$

Out[ ]=

$(-1)^n$

In[ ]:=
```
Simplify[Sqrt[x^2], Assumptions → {x > 0}]
```

Out[ ]=

x

FullSimplify vs Simplify

In[ ]:=
```
Simplify[Gamma[x] x]
```

Out[ ]=

x Gamma[x]

In[ ]:=
```
FullSimplify[Gamma[x] x]
```

Out[ ]=

Gamma[1 + x]

Solve

Solving for a single variable

In[ ]:=
```
Solve[a x^2 + b x + c == 0, x]
```

Out[ ]=

$$\left\{\left\{x \to \frac{-b - \sqrt{b^2 - 4 a c}}{2 a}\right\}, \left\{x \to \frac{-b + \sqrt{b^2 - 4 a c}}{2 a}\right\}\right\}$$

In[ ]:=
```
Solve[a0 + a1 x + a2 x^2 + a3 x^3 + a4 x^4 + a5 x^5 == 0, x]
```

Out[ ]=

```
{{x → Root[a0 + a1 #1 + a2 #1^2 + a3 #1^3 + a4 #1^4 + a5 #1^5 &, 1]},
 {x → Root[a0 + a1 #1 + a2 #1^2 + a3 #1^3 + a4 #1^4 + a5 #1^5 &, 2]},
 {x → Root[a0 + a1 #1 + a2 #1^2 + a3 #1^3 + a4 #1^4 + a5 #1^5 &, 3]},
 {x → Root[a0 + a1 #1 + a2 #1^2 + a3 #1^3 + a4 #1^4 + a5 #1^5 &, 4]},
 {x → Root[a0 + a1 #1 + a2 #1^2 + a3 #1^3 + a4 #1^4 + a5 #1^5 &, 5]}}
```

Equations with multiple solutions

In[ ]:=
```
Show[{
  ContourPlot[x^4 + y^4 == 1, {x, -2, 2}, {y, -2, 2}],
  Plot[3 x, {x, -2, 2}, PlotStyle → ColorData[97, 2]]
 }]
```

Out[ ]=



In[ ]:=
```
Solve[{
  x^4 + y^4 == 1,
  y == 3 x
 }, {x, y}, Reals]

sol = Solve[{
   x^4 + y^4 == 1,
   y == 3 x
  }, {x, y}, Assumptions → {x ∈ Reals, y ∈ Reals}]
```

Out[ ]=

$$\left\{\left\{x \to -\frac{1}{82^{1/4}}, y \to -\frac{3}{82^{1/4}}\right\}, \left\{x \to \frac{1}{82^{1/4}}, y \to \frac{3}{82^{1/4}}\right\}\right\}$$

Out[ ]=

$$\left\{\left\{x \to -\frac{1}{82^{1/4}}, y \to -\frac{3}{82^{1/4}}\right\}, \left\{x \to \frac{1}{82^{1/4}}, y \to \frac{3}{82^{1/4}}\right\}\right\}$$

In[∘]:=
```
{x, y} /. sol⟦2⟧
```
Out[∘]=

$$\left\{\frac{1}{82^{1/4}}, \frac{3}{82^{1/4}}\right\}$$

In[∘]:=
```
Show[{
  ContourPlot[x^4 + y^4 == 1, {x, -2, 2}, {y, -2, 2}],
  Plot[3 x, {x, -2, 2}, PlotStyle → ColorData[97, 2]],
  ListPlot[{x, y} /. sol]
 }]
```
Out[∘]=



In[∘]:=
```
Solve[{
  x^4 + y^4 == a,
  y == b x
 }, {x, y}, Reals]
```
Out[∘]=

$$\left\{\left\{x \to \boxed{-\frac{\left(\frac{a b^4}{1+b^4}\right)^{1/4}}{b} \quad \text{if } a > 0}, y \to \boxed{-\left(\frac{a b^4}{1+b^4}\right)^{1/4} \quad \text{if } a > 0}\right\},\right.$$

$$\left.\left\{x \to \boxed{\frac{\left(\frac{a b^4}{1+b^4}\right)^{1/4}}{b} \quad \text{if } a > 0}, y \to \boxed{\left(\frac{a b^4}{1+b^4}\right)^{1/4} \quad \text{if } a > 0}\right\}\right\}$$

In[∘]:=
```
Solve[{
    x^4 + y^4 == a,
    y == b x
  }, {x, y}, Reals] // Normal
```

Out[∘]=

$$\left\{\left\{x \to -\frac{\left(\frac{a\,b^4}{1+b^4}\right)^{1/4}}{b},\ y \to -\left(\frac{a\,b^4}{1+b^4}\right)^{1/4}\right\},\ \left\{x \to \frac{\left(\frac{a\,b^4}{1+b^4}\right)^{1/4}}{b},\ y \to \left(\frac{a\,b^4}{1+b^4}\right)^{1/4}\right\}\right\}$$

Other useful functions

https://reference.wolfram.com/language/tutorial/AlgebraicCalculations.html

In[∘]:=
```
Factor[2 x^2 + 3 x + 1]
```

Out[∘]=

$(1 + x)\ (1 + 2\,x)$

In[∘]:=
```
Expand[(1 + x) (1 + 2 x)]
```

Out[∘]=

$1 + 3\,x + 2\,x^2$

In[∘]:=

$$\text{Apart}\left[\frac{1}{\left(x^2 + 3\,x + 1\right)\ (3\,x + 1)\ \left(x^2 - 1\right)\ x}\right]$$

Out[∘]=

$$\frac{1}{40\ (-1 + x)} - \frac{1}{x} + \frac{1}{4\ (1 + x)} + \frac{243}{8\ (1 + 3\,x)} + \frac{-123 - 47\,x}{5\ \left(1 + 3\,x + x^2\right)}$$

In[∘]:=
```
TrigExpand[Sin[2 x]]
```

Out[∘]=

$2\,\text{Cos}[x] \times \text{Sin}[x]$